NAG Library Routine Document

G05PFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05PFF generates a given number of terms of a GJR GARCH(p, q) process (see Glosten *et al.* (1993)).

2 Specification

```
SUBROUTINE G05PFF(DIST, NUM, IP, IQ, THETA, GAMMA, DF, HT, ET, FCALL, R,
1                 LR, STATE, IFAIL)
INTEGER           NUM, IP, IQ, DF, LR, STATE(lstate), IFAIL
double precision  THETA(IQ+IP+1), GAMMA, HT(NUM), ET(NUM), R(LR)
LOGICAL           FCALL
CHARACTER*1       DIST
```

3 Description

A GJR GARCH(p, q) process is represented by:

$$h_t = \alpha_0 + \sum_{i=1}^{q}(\alpha_i + \gamma I_{t-i})\epsilon_{t-i}^2 + \sum_{i=1}^{p}\beta_i h_{t-i}, \qquad t = 1, 2, \ldots, T;$$

where $I_t = 1$ if $\epsilon_t < 0$, $I_t = 0$ if $\epsilon_t \geq 0$, and $\epsilon_t \mid \psi_{t-1} = N(0, h_t)$ or $\epsilon_t \mid \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t-distribution with df degrees of freedom and variance h_t, T is the number of observations in the sequence, ϵ_t is the observed value of the GARCH(p, q) process at time t, h_t is the conditional variance at time t, and ψ_t the set of all information up to time t. Symmetric GARCH sequences are generated when γ is zero, otherwise asymmetric GARCH sequences are generated with γ specifying the amount by which negative shocks are to be enhanced.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05PFF.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and Testing the Impact of News on Volatility *Journal of Finance* **48** 1749–1777

Glosten L, Jagannathan R and Runkle D (1993) Relationship between the expected value and the volatility of nominal excess return on stocks *Journal of Finance* **48** 1779–1801

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

1: DIST – CHARACTER*1 *Input*

On entry: the type of distribution to use for ϵ_t.

DIST = 'N'
A Normal distribution is used.

DIST = 'T'
A Student's t-distribution is used.

Constraint: DIST = 'N' or 'T'.

2: NUM – INTEGER *Input*

On entry: T, the number of terms in the sequence.

Constraint: NUM > 0.

3: IP – INTEGER *Input*

On entry: the number of coefficients, β_i, for $i = 1, 2, \ldots, p$.

Constraint: IP ≥ 0.

4: IQ – INTEGER *Input*

On entry: the number of coefficients, α_i, for $i = 1, 2, \ldots, q$.

Constraint: IQ ≥ 1.

5: THETA(IQ + IP + 1) – ***double precision*** array *Input*

On entry: the first element must contain the coefficient α_o, the next IQ elements must contain the coefficients α_i, for $i = 1, 2, \ldots, q$. The remaining IP elements must contain the coefficients β_j, for $j = 1, 2, \ldots, p$.

Constraints:

$$\sum_{i=2}^{\text{IQ+IP+1}} \text{THETA}(i) < 1.0;$$

$$\text{THETA}(i) \geq 0.0, \text{ for } i = \text{IQ} + 2, \ldots, \text{IQ} + \text{IP} + 1.$$

6: GAMMA – ***double precision*** *Input*

On entry: the asymmetry parameter γ for the GARCH(p, q) sequence.

Constraint: GAMMA + THETA$(i) \geq 0.0$, for $i = 2, 3, \ldots, \text{IQ} + 1$.

7: DF – INTEGER *Input*

On entry: the number of degrees of freedom for the Student's t-distribution.

If DIST = 'N', DF is not referenced.

Constraint: if DIST = 'T', DF > 2.

8: HT(NUM) – ***double precision*** array *Output*

On exit: the conditional variances h_t, for $t = 1, 2, \ldots, T$, for the GARCH(p, q) sequence.

9: ET(NUM) – ***double precision*** array *Output*

On exit: the observations ϵ_t, for $t = 1, 2, \ldots, T$, for the GARCH(p, q) sequence.

10: FCALL – LOGICAL *Input*

On entry: if FCALL = .TRUE., a new sequence is to be generated, otherwise a given sequence is to be continued using the information in R.

11: R(LR) – ***double precision*** array *Input/Output*

On entry: the array contains information required to continue a sequence if FCALL = .FALSE..

On exit: contains information that can be used in a subsequent call of G05PFF, with FCALL = .FALSE..

12: LR – INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05PFF is called.

Constraint: LR $\geq 2 \times (\text{IP} + \text{IQ} + 2)$.

13: STATE($lstate$) – INTEGER array *Communication Array*

Note: $lstate$ = LSTATE, the dimension of STATE as supplied to the initialization routine G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

14: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, DIST $\neq$ 'N' or 'T'.

IFAIL = 2

On entry, NUM ≤ 0.

IFAIL = 3

On entry, IP < 0.

IFAIL = 4

On entry, IQ < 1.

IFAIL = 5

On entry, $\alpha_i + \gamma < 0$ for some α_i.

IFAIL $= 7$

On entry, DIST $=$ 'T' and DF ≤ 2.

IFAIL $= 11$

The value of IP or IQ is not the same as when R was set up in a previous call.

IFAIL $= 12$

On entry, LR $< 2 \times (\text{IP} + \text{IQ} + 2)$.

IFAIL $= 13$

On entry, STATE vector was not initialized or has been corrupted.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example first calls G05KFF to initialize a base generator then calls G05PFF to generate two realizations, each consisting of ten observations, from a GJR GARCH$(1, 1)$ model.

9.1 Program Text

```
*     G05PFF Example Program Text
*     Mark 22 Release. NAG Copyright 2006.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          MSTATE, MSEED, NUM, IP, IQ
      PARAMETER        (MSTATE=633,MSEED=1,NUM=10,IP=1,IQ=1)
      INTEGER          LR
      PARAMETER        (LR=2*(IQ+IP+2))
*     .. Local Scalars ..
      DOUBLE PRECISION GAMMA
      INTEGER          DF, GENID, I, IFAIL, LSEED, LSTATE, SUBID
      LOGICAL          FCALL
      CHARACTER        DIST
*     .. Local Arrays ..
      DOUBLE PRECISION ET(NUM), HT(NUM), R(LR), THETA(IP+IQ+1)
      INTEGER          SEED(MSEED), STATE(MSTATE)
*     .. External Subroutines ..
      EXTERNAL         G05KFF, G05PFF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05PFF Example Program Results'
      WRITE (NOUT,*)
*     Initialize the seed
      SEED(1) = 1762543
*     GENID and SUBID identify the base generator
      GENID = 1
      SUBID = 1
*     Initialize the generator to a repeatable sequence
      LSTATE = MSTATE
      LSEED = MSEED
      IFAIL = 1
      CALL G05KFF(GENID,SUBID,SEED,LSEED,STATE,LSTATE,IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99998) IFAIL
         GO TO 60
```

```
      END IF
*     Set up the parameters for the series being generated
      DIST = 'N'
      THETA(1) = 0.4D0
      THETA(2) = 0.1D0
      THETA(3) = 0.7D0
      GAMMA = 0.1D0
      DF = 0
      FCALL = .TRUE.
*     Generate the first realisation
      IFAIL = 1
      CALL G05PFF(DIST,NUM,IP,IQ,THETA,GAMMA,DF,HT,ET,FCALL,R,LR,STATE,
     +            IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 60
      END IF
*     Display the results
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Realisation Number 1'
      WRITE (NOUT,*) '   I            HT(I)            ET(I)'
      WRITE (NOUT,*) '  ---------------------------------------'
      DO 20 I = 1, NUM
         WRITE (NOUT,99999) I, HT(I), ET(I)
   20 CONTINUE
*     Generate a second realisation
      FCALL = .FALSE.
      IFAIL = 1
      CALL G05PFF(DIST,NUM,IP,IQ,THETA,GAMMA,DF,HT,ET,FCALL,R,LR,STATE,
     +            IFAIL)
      IF (IFAIL.NE.0) THEN
         WRITE (NOUT,99997) IFAIL
         GO TO 60
      END IF
*     Display the results
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Realisation Number 2'
      WRITE (NOUT,*) '   I            HT(I)            ET(I)'
      WRITE (NOUT,*) '  ---------------------------------------'
      DO 40 I = 1, NUM
         WRITE (NOUT,99999) I, HT(I), ET(I)
   40 CONTINUE
*
   60 CONTINUE
*
99999 FORMAT (1X,I5,2(1X,F16.4))
99998 FORMAT (1X,' ** G05KFF returned with IFAIL = ',I5)
99997 FORMAT (1X,' ** G05PFF returned with IFAIL = ',I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05PFF Example Program Results


  Realisation Number 1
    I            HT(I)            ET(I)
   ---------------------------------------
     1          1.8000           0.4679
     2          1.6819          -1.6152
     3          2.0991           0.9592
     4          1.9614           1.1701
     5          1.9099          -1.7355
     6          2.3393          -0.0289
     7          2.0377          -0.4201
     8          1.8617           1.0865
```

```
     9           1.8212          -0.0061
    10           1.6749           0.5754

  Realisation Number 2
    I            HT(I)            ET(I)
   -------------------------------------
     1           1.6055          -2.0776
     2           2.3872          -1.0034
     3           2.2724           0.4756
     4           2.0133          -2.2871
     5           2.8554           0.4012
     6           2.4149          -0.9125
     7           2.2570          -1.0732
     8           2.2102           3.7105
     9           3.3239           2.3530
    10           3.2804           0.1388
```